

Special Symbols in Graphs: Multiple Solutions

Abhinav Srivastva, Gilead Sciences

ABSTRACT

It is not uncommon in Graphs to include special symbols at various places like axes, legends, titles and footnotes, and practically anywhere in the plot area. The paper discusses multiple ways how special symbols can be inserted as applicable in SAS/GRAPH®, Graph Annotations, ODS Graphics® - SG Procedures, SG Annotations and Graph Template Language (GTL). There will be some examples presented which leverage the power of Formats to put these into action. The techniques will vary depending on the version of SAS® and the type of procedure (PROC) used.

INTRODUCTION

Graphs are a routine part of FDA submission like Clinical Study Report (CSR), NDA/BLA in depicting key efficacy and safety measures. With the increased need and complexity, there often present challenges to display information in a way that facilitates readability to the consumer.

SAS enables adding special symbols in graphs through traditional to more advanced ways as presented in the paper. The techniques include – copy/paste from Character map, keystroke sequences (Alt+XXXX), changing font style, incorporating hexadecimal characters, using byte function, with move=<option>, In-line formatting, unicode characters in combination with escape character and attribute mapping in ODS Graphics. In addition, SYMBOLCHAR and SYMBOLIMAGE in SG allows symbols and images to be inserted in a unique way.

OVERVIEW OF SOLUTIONS

A brief discussion along with a collection of examples have been selected to demonstrate the ways by which special symbols can be included in graphs.

- Method 1: Copy/paste from Character map
- Method 2: Keystroke sequences (Alt+XXXX)
- Method 3: Changing Font Style
- Method 4: Hexadecimal characters
- Method 5: BYTE function
- Method 6: MOVE=<option>
- Method 7: In-line formatting
- Method 8: Unicode characters
- Method 9: Attribute mapping in SG Procedures and GTL
- Method 10: SYMBOLCHAR and SYMBOLIMAGE in SAS 9.4

METHOD 1: COPY/PASTE FROM CHARACTER MAP

Windows provide a system character map (Start ➔ Accessories ➔ System tools ➔ Character map) where symbols can be copied and pasted directly into the SAS editor. The default font style that appears in Character map is 'Arial' but that can be changed from the drop-down list to select desired symbols available within a given font style. While it's a quick and easy way to insert symbols, some symbols don't embed well into the editor and need other alternatives to handle those special cases.

	<pre> proc sgplot data=sashelp.class; scatter x=height y=weight/group=sex; title "Method 1: Using Character map"; title2 "Inserting dagger (†) symbol"; run; </pre>

Display 1: Copy/Paste from Character map to add special symbols

METHOD 2: KEYSTROKE SEQUENCES

As an extension of Method 1: copy/paste from character map, keystroke sequence that get displayed at the bottom of the character map for a selected character can be entered in the SAS editor to render that character. For the above example, (ALT+0134) will produce a dagger symbol (†) in the output.

METHOD 3: CHANGING FONT STYLE

SAS has an extensive list of font styles shipped with the product. The following code prints the list of fonts stored in the catalog. From the collection of font styles, any font style can be selected ('greek' in this example) to know the keyboard character associated with corresponding special symbol character.

<pre> proc catalog catalog=sashelp.fonts entrytype=font; contents out=work.swfonts (keep=name); run; quit; data work.swfonts; set work.swfonts; if name =:'HW' then delete; run; proc print data=work.swfonts; run; </pre>	<pre> options reset=all border hsz=5.5in vsize=4.14in device=png; proc gfont name=greek nobuild height=1.5 romcol=red romfont=swiss romht=1 showroman; run; </pre>
--	---

The below program uses 'greek', 'special' and 'simplex' font to display special symbols at various places in the plot. Annotate facility has been used to add name text 'Alfred' corresponding to its height & weight data.

```

symbol1 i=none f=marker v=U c=CXDE7E6F;
symbol2 i=none f=marker v=U c=CX7C95CA;


title1 'Method 3: Displaying special symbols with SAS/GRAPH output';
title2 font='Albany AMT' 'Height  $\Delta$  Weight' font='greek' 'DV' font='Albany AMT'
'Weight';
axis1 label=(font='greek' 'w' font='Albany AMT' 'eight');

data anno;
  set sashelp.class (where=(name='Alfred'));
  length text $10;
  xsys='2'; ysys='2'; when='a';
  function='LABEL'; style='SPECIAL'; position='2';
  x=weight; y=height;
  text='M';
  output;
  function='MOVE';
  x=x+2; y=y+2;
  output;
  function='LABEL'; style='SIMPLEX'; position='6';
  text=strip(name);
  output;

run;

proc gplot data=sashelp.class;
  plot height*weight=sex / anno=anno haxis=axis1;
run;
quit;

```




Display 2: Changing Font style with SAS/GRAPH to add special symbols

METHOD 4: USING HEXADECIMAL CHARACTERS

Hexadecimal characters can be generated from ASCII characters using \$HEX format. The below DATA step demonstrates this procedure. Please note: ASCII characters are operating system-sensitive so not all the hexadecimal characters can be applied correctly within the SAS programs.

```
data hex_table;
  do ascii=0 to 255;
    char=byte(ascii);
    hex=put(char,$HEX.);
    output;
  end;
run;
```

The hexadecimal characters can be incorporated in Method 3: Changing font style, with SAS/GRAPH to produce the same effect. The method is independent of the font style and can be expanded to SG Procedures as well in a very limited role compared to 'unicode' characters which provide a more holistic approach to add special symbols in ODS Graphics. For completeness, below is an example where a 2-digit hexadecimal character ('£' symbol) is added on Y-axis using SG PLOT.




Display 3: Using Hexadecimal characters to add special symbols

METHOD 5: USING BYTE FUNCTION

To complement hexadecimal characters, byte function, BYTE(n), can render special symbols from the ASCII list (0-255). Though based on the operating system, some of the ASCII characters might not be available for use. Byte function can be implemented directly in the SAS statements or stored in a macro available to be used anywhere in the program. The above example is enhanced using BYTE function to add more description on Y-axis label as below which could not be accomplished in the previous example.

```
data _null_;
  call symputx('char',byte(163));
run;

proc sgplot data=sashelp.prdsale (where=(country='CANADA'));
  vbar product / response=actual stat=mean barwidth=0.4;
  title "Method 5: Byte function in SG PLOT";
  title2 "Example produces (%sysfunc(byte(163))) symbol";
  xaxis label='Product';
  yaxis label="Amount in &char." ;
run;
```




Display 4: Using BYTE function to add special symbols

METHOD 6: WITH MOVE=<OPTION>

Move=(±m,±n) can be used in TITLE, FOOTNOTE and NOTE statements to move text which can even facilitate adding special symbols when used in combination with Method 3 through Method 5. The below example demonstrates the application of MOVE option.

```
proc gplot data=sashelp.class;
  plot height*weight=sex / haxis=axis1;
  title "Method 6: Move=<option> in GPLOT";
  title3 "M vs F" move=(+0.5,+0.5) h=1.1 "%sysfunc(byte(167))";
  footnote " " move=(0,+0.5) f=special "N" move=(+0.5,-0.5)
    font="Albany AMT" "M=Male, F=Female";
run;
```



Display 5: With Move=<option> to add special symbols

METHOD 7: IN-LINE FORMATTING


In-line formatting of titles, footnotes, data and text fields can be achieved through the use of ODS ESCAPECHAR. Some of the features include adding pre and post-images, special symbols, super and subscripts and other text formatting features. The below example leverages some of the in-line formatting capabilities to exhibit its usefulness in graphs.

```
ods escapechar='^';
title "In-line formatting Example";
title2 color=red 'This is a plot by Height & Weight^{dagger}';
title3 '^S={preimage="sashelp_data.png"}';

ods pdf file="&path.\method7.pdf" style=journal nogtitle nogfootnote;
-----
proc gplot data=sashelp.class;
  plot height*weight=sex / haxis=axis1;
run;
quit;

----- OR -----

proc sgplot data=sashelp.class;
  scatter x=weight y=height / group=sex;
run;
ods pdf close;
```



Display 6: In-line formatting to add special symbols

METHOD 8: USING UNICODE CHARACTERS

Unicode characters represent the most powerful way to add special symbols in ODS Graphics. The character map as seen in Display 1: Copy/Paste from Character map to add special symbols, displays the 4-digit Unicode character for a selected symbol at the bottom of the dialogue box. The syntax is $\wedge\{\text{Unicode xxxx}\}$ where \wedge is the specified ODS ESCAPECHAR and xxxx is the 4-digit Unicode value. The method can be executed in a number of ways.

Program 8a: Using Format to Add Unicode characters in Axis values, Labels and Titles

```

proc format;
  value $sex
    'F' = "(*ESC*){unicode '2640'x}"
    'M' = "(*ESC*){unicode '2642'x}"
  ;
run;

title "Method 8: Unicode Character, Example 1";
title2 "Male ^{unicode '2194'x} Female>";

proc sgplot data=sashelp.class;
  format sex $sex.;
  vbox height / category=sex;
  yaxis label = "(*ESC*){unicode '25BA'x} height";
  inset "(*ESC*){unicode '00AB'x} Works in Inset too"/
  position=bottomright;
run;

/*----- Axis values as above can also be obtained with SAS GTL -----*/
proc template;
  .....
  drawtext {unicode '2640'x} /xspace=datavalue yspace=datavalue y=-4 x='F';
  drawtext {unicode '2642'x} /xspace=datavalue yspace=datavalue y=-4 x='M';
  ...
run;

```



Program 8b: Adding Unicode characters in Legend

```

data class;
  set sashelp.class;
  if sex='M' then do;
    ht1=height; wt1=weight;
  end;
  if sex='F' then do;
    ht2=height; wt2=weight;
  end;
run;

title "Method 8: Unicode Character, Example 2";
title2 "Male ^{unicode '2194'x} Female>";
proc sgplot data=class;
  scatter x=wt1 y=ht1 / legendlabel="Male ((*ESC*){unicode '2642'x})";
  scatter x=wt2 y=ht2 / legendlabel="Female ((*ESC*){unicode '2640'x})";
  keylegend / title=' ' location=inside position=topleft across=1;
  xaxis label="Weight";
  yaxis label="Height";
run;

```



Display 7: Outputs from program 8a and 8b; Unicode characters to add special symbols

METHOD 9: ATTRIBUTE MAPPING IN SG PROCEDURES AND GTL

Attribute mapping allows marker symbols and other attributes to be managed through the creation of attribute dataset in a DATA step. Some of the common attributes that are of interest to the user are 'markercolor', 'markersymbol', 'linecolor', 'linepattern' and 'fillcolor' along with required ID and VALUE

METHOD 10: SYMBOLCHAR AND SYMBOLIMAGE IN SAS 9.4


To expand the list of marker symbols already available in SAS; SYMBOLCHAR and SYMBOLIMAGE beginning in SAS 9.4 allows special characters and images to be placed as the markers. In SYMBOLCHAR a character from any font style any be listed while in SYMBOLIMAGE an external image can be loaded as markers. Below examples put these statements into action.


```

----- SYMBOLCHAR -----
proc sgplot data= sashelp.class;
  symbolchar name=male char='2642'x / textattrs=(family='Arial Unicode MS'
                                                weight=bold);
  symbolchar name=female char='2640'x / textattrs=(family='Arial Unicode
                                                MS' weight=bold);
  styleattrs datasymbols=(male female);
  scatter y=height x=weight / group=sex name='a' markerattrs=(color=black
                                                            size=30) dataskin=sheen;
  keylegend / location=inside position=topleft opaque;
  xaxis grid offsetmin=0.05 offsetmax=0.05;
  yaxis grid offsetmin=0.05 offsetmax=0.05;;
run;

----- SYMBOLIMAGE -----
proc sgplot data= sashelp.class;
  symbolimage name=male image="&path.\male.png";
  symbolimage name=female image="&path.\female.png";
  styleattrs datasymbols=(male female);
  scatter y=height x=weight / group=sex name='a' markerattrs=(color=black
                                                            size=30);
  keylegend / location=inside position=topleft opaque;
  xaxis grid offsetmin=0.05 offsetmax=0.05;
  yaxis grid offsetmin=0.05 offsetmax=0.05;;
run;

```





Display 9: SYMBOLCHAR and SYMBOLIMAGE to add special symbols

CONCLUSION

As seen from the above examples, there are multiple ways to generate special symbols depending on the version of SAS and the type of procedure (PROC) used. Also, as noticeable through the progression of the methods presented, the earlier methods are more suited for SAS/GRAPH (GHCART, GPLOT etc) and the latter methods for ODS Graphics (SGPLOT, SGSCATTER etc). Based on feasibility of the objects (axes, legends, marker symbols, titles etc) that are allowed to be modified through the methods, my recommendation is to use 'Font style' method (METHOD 3) when dealing with SAS/GRAPH and 'Unicode' method (METHOD 8) when dealing with ODS Graphics as those enable more manipulating power than their counterparts.

REFERENCES

SAS/GRAPH® 9.4 Reference, Fifth Edition, Cary NC.
<http://support.sas.com/documentation/cdl/en/graphref/69717/HTML/default/viewer.htm#n1kmdfggzrjeokn1xd17pvzxg2bi.htm>

SAS® 9.4 ODS Graphics: Procedures Guide, Sixth Edition, Cary NC.
<http://support.sas.com/documentation/cdl/en/grstatproc/69716/HTML/default/viewer.htm#p0i3rles1y5mvs n1hrq3i2271rmi.htm>

SAS Support, "Sample 47801: Use special symbols with SAS/GRAPH® output",
<http://support.sas.com/kb/47/801.html>

Matange, Sanjay. Graphically speaking blog "Marker Symbols", Jan 2015,
<http://blogs.sas.com/content/graphicallyspeaking/2015/01/14/marker-symbols/>

Matange, Sanjay. Graphically speaking blog "Displaying Unicode Symbols in Legend", Jan 2015,
<http://blogs.sas.com/content/graphicallyspeaking/2015/01/19/displaying-unicode-symbols-in-legend/>

Matange, Sanjay. Graphically speaking blog "More Symbols, you say?", Jan 2014,
<http://blogs.sas.com/content/graphicallyspeaking/2014/01/26/more-symbols-you-say/>

McConnell, Lelia. 2013. "GTL to the Rescue!", SAS Global Forum 2013 – Paper 382-2013.

McConnell Lelia, SAS blog “Specifying Unicode values and colors easier with PROC SGPLOT and GTL”, Sep 2015, <http://blogs.sas.com/content/sgf/2015/09/18/specifying-unicode-values-and-colors-is-easier-with-the-proc-sgplot-and-gtl/>

McConnell Lelia, SAS blog “Symbols in SAS 9.4 graphs: unlimited possibilities”, Oct 2014, <http://blogs.sas.com/content/sgf/2014/10/17/symbols-in-sas-9-4-graphs-unlimited-possibilities/>

Karidi Kiran, Vanam Mahipal, Dodlapati Sridhar. 2010. “Embedded Special Characters”, PharmaSUG2010 – Paper CC19.

Lassman, Doug. 2003. “Fonts and Special Characters in SAS – It’s all Greek to Me”, NESSUG 16 – Paper PS005.

Hadden, Louise S. 2010. “The Great Escape(char)”, SAS Global Forum 2010 – Paper 215-2010.

Heath Dan, 2014. “Putting on the Ritz: New Ways to Style Your ODS Graphics to the Max ”, MWSUG2014 – Paper SS02.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Abhinav Srivastva
Enterprise: Gilead Sciences, Inc.
E-mail: srivastvaabhinav@gmail.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.